объединенный
институт
ядерных
исследований
дубна

E2-94-110

R.M.Muradian[1*], A.L.Urintsev[2]

DIANA: A MATHEMATICA CODE FOR MAKING DIMENSIONAL ANALYSIS

Submitted to «Mathematica in Education»

[1]Electronic address: muradian@theor.jinrc.dubna.su.
*Permanent address: Byurakan Astrophysical Observatory, Armenia.
[2]Electronic address: urintsev@sunse.jinr.dubna.su.

1994

1 Start Up

Mathematica®[1] is one of the newest and most powerful computer systems for doing symbolic, numerical, graphics, and programming manipulations [1], [2]. With its large capabilities, *Mathematica* opens new prospects for solving a broad range of problems in physics, engineering and economics.

The *Mathematica* package *Diana* was designed and implemented by us for making automatic and quick dimensional analysis for any problem in physics and engineering. The package is based on the fundamental principles of dimensional analysis formulated in the matrix form [3] . Materials from other common textbooks and articles [4]-[12] on dimensional analysis are used. A comprehensive treatment of the modern state of art together with the thorough list of the literature can be found in [12]. The basic principles of dimensional analysis are realized by using capabilities provided by the *Mathematica*'s functions `LinearSolve` and `NullSpace`[1].

It is proposed to increase the number of the SI base units by addition of the new fundamental unit of price — dollar $ — for solving financial problems in engineering physics.

A detailed SI Units Reference Table is enclosed at the subsection 4.4 for user's convenience. Table was compiled on the basis of the [13]–[15].

2 User's Guide

This section describes the loading of the package into *Matematica* and the standard help-mechanism. This information together with the information on SI units collected in subsection 4.4 are sufficient for practical applications of the package in different areas of science .

[1] *Mathematica* is a registered trademark of Wolfram Research, Inc.

2.1 Getting

The file *Diana*.m and its documentation are available by electronic mail from the authors.

2.2 Loading

It is assumed that the package is located in the *Mathematica*-path, e.g. in user's current directory. Then the package is loaded into a *Mathematica* session using one of the the commands `Needs["Diana'"]`, `Get["Diana'"]`, `<<Diana.m` or `<<Diana'`:

```
In[1]:= <<Diana'

Diana:  Version 1.0, March 1994
        by R. Muradian (1) & A. Urintsev (2)
        (1) Electronic address:  muradian@theor.jinrc.dubna.su
        (2) Electronic address:  urintsev@sunse.jinr.dubna.su
        Type ?Diana'*  for all exported symbols and
        ?Diana,  ?DIANA,  ?toFU  for help on Diana package.
```

The message shows the version of *Diana*, the date of creation and contains information on help-mechanism. Our computations used *Mathematica* 2.2 for MS-Windows.

2.3 Symbols

Using the standard *Mathematica* syntax it is straightforward to obtain a complete listing of all exported symbols defined in the *Diana* by the command `Names["Diana'*"]` or `?Diana'*`:


```
In[2]:= ?Diana'*
$           Diana      Gray       Kilogram   Newton     Siemens     toFU
Ampere      DIANA      Henry      Lumen      Ohm        Sievert     Volt
Becquerel   F          Hertz      Lux        Pascal     Steradian   Watt
Candela     Farad      Joule      Meter      Radian     Tesla       Weber
Coulomb     G          Kelvin     Mole
```

More detailed information is available on any of these symbols, e.g.:

```
In[2]:= ?Diana
Diana[{{name1, SIunit1}, {name2, SIunit2}, ...}] performs
 dimensional analysis and expresses outcome in the form:
 name1 -> expr F(K2, K3, ..., Kn). Here K1, K2, ..., Kn
 (with K1 = name1/expr) represent the complete set of
 dimensionless parameters and F is undetermined function.
 Parameters K2, ..., Kn and expr do not depend on name1.
 If n = 1, the result is  name1 -> expr.
In[2]:= ?DIANA
DIANA[{{name1, SIunit1}, {name2, SIunit2}, ...}] performs
 dimensional analysis and expresses outcome in the form:
 G(K1, K2, ..., Kn) -> 0, where Ki represents the complete set
 of dimensionless parameters and G is undetermined function.
In[2]:= ?toFU
toFU[SIunit] transforms any combination of fundamental and
 derived SI units into fundamental SI units.
In[2]:= ?Farad
 Farad is the derived SI unit of capacitance.
 Farad = (Ampere^2*Second^4)/(Kilogram*Meter^2)
```

3 Illustrative Examples

The following typical examples illustrate how one uses the package and what sorts of problems can be solved with *Diana*.

3.1 Simple Example

Find the volume of the d-dimensional ball:

```
In[2]:= Diana[{{V, Meter^d}, {r, Meter}, {d, 1}}]
                d
Out[2]= V -> r  F[d]
```

The dimensionless factor F[d] cannot be attained by the dimensional analysis method. Its exact form is known from more detailed solution as $(\pi)^{d/2}/\Gamma(1 + d/2)$. DIANA gives another form for the solution:

```
In[3]:= DIANA[{{V, Meter^d}, {r, Meter}}]
          r
Out[3] = G[----] -> 0
           1/d
          V
```

3.2 Vibration of a Star [3]

The following quantities:

1. frequency of vibration (f), 2. mass density (ρ), 3. radius (r), 4. gravitational constant (G)

characterize the system. The solution first obtained by Lord Rayleigh is straightforward:

```
In[4]:= Diana[{{f, Hertz}, {rho, Kilogram/Meter^3}, {r, Meter},
              {G, Newton Meter^2 Kilogram^-2}}]
Out[4]= f -> Sqrt[G] Sqrt[rho]
```

Dimensional analysis shows that the radius is redundant.

3.3 Ship Propeller [3]

The following quantities characterize the system:

1. thrust force of the propeller (f), 2. radius of the propeller (r), 3. density of water (ρ), 4.speed of the ship (V), 5. acceleration due to gravity (g), 6. viscosity of the water (ν), 7. rotational speed of the propeller (ω).

```
In[5]:= Diana[{{f, Newton}, {r, Meter},{rho, Kilogram/Meter^3},
        {V, Meter/Second}, {g, Meter/Second^2},
        {nu, Meter^2/Second}, {omega, Radian/Second}}]

                 2       2   nu    g r   omega r
Out[5]= f -> r  rho V  F[---,  ---,  -------]
                          r V    2        V
                                V
```

The famous dimensionless combinations, the Froude number $\mathbf{Fr} = \mathrm{V}^2/\mathrm{gr}$ and Reynolds number $\mathbf{Re} = \mathrm{Vr}/\nu$, are obtained automatically.

3.4 Airplane Flying Through Rainstorm [3]

Input data:

1. number of raindrops striking the windshield per second (n), 2. characteristic length of the airplane (L), 3. diameter of a raindrop (d), 4. speed of airplane (V), 5. number density of raindrops (N_r), 6. mass density of water (ρ_W), 7. mass density of air (ρ_A), 8. surface tension of water (σ), 9. viscosity of air (μ_a), 10. acceleration due to gravity (g).

Outcome for the number of raindrops:

```
In[6]:= Diana[{{n, 1/Second}, {V, Meter/Second}, {N, Meter^-3},
              {L, Meter},{rhoW, Kilogram/Meter^3},
```

```
          {rhoA, Kilogram/Meter^3}, {sigma, Newton/Meter},
     {g, Meter/Second^2}, {d, Meter}, {mu, Pascal Second}}]
                                1/3            1/3
          1/3         g       N     sigma   mu N      rhoA     1/3
Out[6]= n -> N     V F[-------, ----------, -------, ----, d N    ,
                        1/3  2          2     rhoW V   rhoW
                       N     V    rhoW V
                          1/3
                       L N    ]
```

The function DIANA gives another set of dimensionless combinations:

```
In[7]:= DIANA[{{n, 1/Second}, {V, Meter/Second}, {N, Meter^-3},
              {L,Meter}, {rhoW, Kilogram/Meter^3},
              {rhoA, Kilogram/Meter^3},{sigma, Newton/Meter},
        {g, Meter/Second^2}, {d, Meter}, {mu, Pascal Second}}]
          3
         N V    g   rhoA  n sigma   mu n    d n  L n
Out[7] = G[----, ---, ----, -------, -------, ---, ---] -> 0
           3    n V  rhoW       3        2    V    V
          n                 rhoW V   rhoW V
```

3.5 Atomic Explosion [5], [6]

A large amount of energy is suddenly released in infinitely confined space. The motion of the surrounding air is characterized by:

1. radius of a spherical shock front (r), 2. time since the explosion started (t), 3. the atmospheric density (ρ), 4. the released energy (e).

The solution for r can be found:

```
In[8]:= Diana[{{r, Meter}, {e, Joule}, {rho, Kilogram/Meter^3},
              {t, Second}}]
```



```
                  1/5  2/5
                 e     t
Out[8]= r -> ---------
                  1/5
                 rho
```

The solution for the speed of the shock front (c) is the following:

```
In[9]:= Diana[{{c, Meter/Second}, {e, Joule},
              {rho, Kilogram/Meter^3}, {t, Second}}]
                   1/5
                  e
Out[9]= c -> ----------
                  1/5  3/5
                rho    t
```

It is easy to check that `r = c t`.

3.6 Financial Scaling

Lord Kelvin was interested in the following problem: what cross section must have the conductor to minimize the cost of electric line? The smaller cross section is advantageous from the point of view of the cost of the conductor,but not for the energy lost. The economic optimum can be reached by a balance between these rival tendencies. To solve this problem by the method of dimensional analysis, let us use extended SI units, supplemented by the fundamental unit of price — dollar \$. The new $SI_\$$ units have the following set of the eight fundamental units:

```
{Meter, Kilogram, Second, Ampere, Kelvin, Mole,Candela, $}
```

The list of essential variables in Kelvin's problem is:

1. cross section of the wire (S). 2. the price of the wire (a). 3. the price of energy (b), 4. electric current (i). 5. period of exploitation (t), 6. conductivity (ρ),

Then:

```
In[10]:= Diana[{{S, Meter^2}, {a, $/Meter}, {b, $/Joule},
               {i, Ampere}, {t, Second}, {rho, Ohm Meter}}]
                 2
              b i  rho t
Out[10]= S -> ----------
                 a
```

The solution exhibits an interesting property : it is invariant under scale transformation of the price unit \$ -> λ \$. In other words, the rate of inflation does not affect the optimal cross section of the conductor. The phenomenon of scaling is well known from different branches of physics [8], [9]. Its generalization for the problems in economics will be quite useful.

3.7 Schrödinger Equation [10]

Consider the quantum-mechanical problem of determining the energy levels in the potential $V(x) = g\,r^k$. The relevant physical quantities are :

1. energy (e), 2. coupling constant (g), 3. Planck constant (h), 4. mass of a particle (m), 5. exponent (k).

The dimension of g can be obtained from the condition that $g\,r^k$ is energy. Dimensional analysis gives the following factor for the energy:

```
In[11]:= Diana[{{e, Joule}, {h, Joule Second}, {g, Joule/Meter^k},
               {m, Kilogram}, {k,1}]
```



```
                2/(2 + k)   (2 k)/(2 + k)
               g           h                  F[k]
Out[11]= e -> -------------------------------
                      k/(2 + k)
                     m
```

Since the quantum number n appears only in combination nh, we can obtain quantum-number dependence of energy by substitution h -> nh:
e_n -> $n^{2k/(k+2)}$.

3.8 Classical Electron Radius [14]

The following quantities determine the problem:

1. classical electron radius (r), 2. elementary charge (e), 3. permittivity of free space (ϵ), 4. speed of light (c), 5. electron mass (m).

The solution is:

```
In[12]:= Diana[{{r, Meter}, {e, Coulomb}, {eps, Farad/Meter},
                {m, Kilogram}, {c, Meter/Second}}]
                  2
                 e
Out[12]= r -> --------
                2
               c  eps m
```

3.9 Stefan Constant [14]

Input data:

1. Stefan constant (σ), 2. Boltzmann constant (k), 3. Planck constant (h) , 4. speed of light (c).

Outcome:

```
In[13]:= Diana[{{sigma, Watt Meter^-2 Kelvin^-4},
            {k, Joule/Kelvin},{h, Joule Second},
            {c, Meter/Second}}]
                4
               k
Out[13]= sigma -> -----
              2  3
             c  h
```

3.10 Clapeyron Equation [14]

Input data:

> 1. molar volume (V), 2. molar gas constant (R), 3. temperature (T), 4. pressure (p).

Outcome:

```
           R T
Out[14]= V -> ---
            P
```

3.11 Exposure Rate [15]

Input data:

> 1. exposure-rate (ExposureRate), 2. activity (A), 3. distance from a radioactive source (L), 4. exposure rate constant (Γ).

Outcome:

```
In[15]:= Diana[{{ExposureRate, Coulomb/(Kilogram Second)},
           {A, Becquerel}, {Gamma, Becquerel^-1 Meter^2
           Coulomb Kilogram^-1 Second^-1}, {L, Meter}}]
```



```
                              A Gamma
Out[15]= ExposureRate -> -------
                                 2
                                L
```

3.12 Checking Equations

The toFU function can be used for checking correctness of the physical equations. For example, in the presence of magnetic field the commutator of the momentum operators obeys the following 'Landau quantization' rule:

$$[p_1, p_2] = i e\, B_3\, \hbar$$

Application of the toFU function to each side proves the dimensional correctness of this equation:

```
In[16]:= toFU[(Newton Second)^2]
                  2        2
         Kilogram  Meter
Out[16]= ----------------
                     2
               Second
In[17]:= toFU[Coulomb Tesla Joule Second]
                  2        2
         Kilogram  Meter
Out[17]= ----------------
                     2
               Second
```

4 Basics

Dimensional analysis consists of two steps :

1. choice of relevant variables, 2. construction of a complete set of dimensionless combinations from these variables.

The first step requires a deep physical insight into essence of the problem and is beyond the scope of ability of the computers. The second step can be made by means of high-level symbolic computer algebra systems, and our *Mathematica*'s code *Diana* provides a possible solution.

The choice of the SI units for manipulation does not restrict by no means the generality of the method.

4.1 SI Units: a Reminder

The International System of units (SI) is based on the seven fundamental units of length, mass, time, electric current, (thermodynamic) temperature, amount of substance, and luminous intensity. The derived dimensions are introduced for the sake of efficiency.

The important property of SI Units, which make these convenient for use in dimensional analysis method, is their coherence. The system of units is said to be coherent when derived units are expressed in terms of the fundamental units with numerical factors equal to unity. The second advantage of SI Units is their adoption throughout the world. Information on SI Units necessary for our purposes is collected in subsection 4.4.

Certain *Mathematica*'s standard packages : Miscellaneous'SIUnits', Miscellaneous'Units' and Miscellaneous'PhysicalConstants' [2] operates on the same physical quantities as *Diana*. The acquaintance with these packages would be desirable but not compulsory.

4.2 $SI_\$$ Units

The method of dimensional analysis can be extended upon inclusion into a set of fundamental SI units the unit of price – dollar \$. These extended units are defined in the following way: meter, kilogram, second, Ampere, Kelvin, mole, candela,dollar. In *Mathematica* they are: `Meter, Kilogram, Second,`

`Ampere, Kelvin, Mole, Candela, $`. This approach enables one to apply the power of dimensional analysis method to the solution of the engineering economic problems. The Example 5.6 demonstrate this possibility. Other applications can be found in designing large-scale engineering systems. It seems appropriate to use the symbol $SI_\$$ for this extended system of units.

4.3 Buckingham–Riabouchinsky Theorem

As a matter of fact, all dimensional analysis is contained in the Buckingham–Riabouchinsky theorem[2]. In a modern treatment [3], the theorem states that the number n of dimensionless complexes is equal to:

$$n = number\, of\, variables\ - rang\, of\, the\, dimensional\, matrix.$$

Thus, the mathematical law describing a physical phenomenon can be expressed as:

$$G(K_1, K_2, ..., K_n) = 0.$$

This equation easily gives :

$$K_1 = F_1(K_2, K_3, ..., K_n),$$

or

$$K_2 = F_2(K_1, K_3, ..., K_n),$$

etc. If $n = 1$, then $K_1 = const.$

The explicit form of the functions G and F_i $(i = 1, ..., n)$ remains undetermined in the framework of the dimensional analysis. This is the major limitation of the method. The second one is the "foresight" required to choose the right complete set of variables and dimensional constants characterizing correctly the problem.

[2]Russian scientist D.P.Riabouchinsky is well known as a author of the famous Rayleigh-Riabouchinsky paradox. Buckingham himself gives the credit for general theorem of the dimensional analysis to Riabouchinsky [4, page 42]. The name π-theorem also is adopted instead of Buckingham–Riabouchinsky theorem.

4.4 SI Units Reference Table

Warning 1. Do not use protected symbol I as a name in input data.

Warning 2. *Diana* makes use of the system protected symbol Second.

Name of Quantity	*SI units for Physics*	*SI Units for Mathematica*
length (r,d,l)	m	`Meter`
mass (m)	kg	`Kilogram`
time (t)	s	`Second`
electric current (i)	A	`Ampere`
temperature (T)	K	`Kelvin`
amount of substance (n)	mol	`Mole`
luminous intensity (I_v)	cd	`Candela`
price in dollars(Pr)	$	`$`
plane angle ($\alpha, \beta, \gamma, \theta, \phi$)	rad=m/m	`Radian`
solid angle (Ω, ω)	$sr=m^2/m^2$	`Steradian`
frequency (f, ν)	$Hz=s^{-1}$	`Hertz`
force (F)	$N=m\ kg\ s^{-2}$	`Newton`
pressure (p)	$Pa=N/m^2$	`Pascal`
energy (E,W)	J=N m	`Joule`
power (P)	W =J/s	`Watt`
speed (v,c)	$m\ s^{-1}$	`Meter/Second`
acceleration (a,g)	$m\ s^{-2}$	`Meter Second^-2`
momentum (p)	N s	`Newton Second`
angular momentum (L,J)	J s	`Joule Second`
dynamic viscosity (μ)	Pa s	`Pascal Second`
kinematic viscosity (ν)	$m^2\ s^{-1}$	`Meter^2/Second`
surface tension (σ)	$N\ m^{-1}$	`Newton/Meter`
modulus of elasticity (Y,E)	$N\ m^{-2}$	`Newton Meter^-2`

gravitational constant (G_N)	$N\ m^2\ kg^{-2}$	`Newton Meter^2 Kilogram^-2`
electric charge (e,q)	C=A s	`Coulomb`
electric potential(ϕ,U,V)	V=W/A	`Volt`
capacitance (C)	F=C/V	`Farad`
resistance (R)	$\Omega=V/A$	`Ohm`
conductance (G)	$S=A/V=\Omega^{-1}$	`Siemens`
resistivity (ρ)	Ω m	`Ohm Meter`
conductivity (σ)	S/m	`Siemens/Meter`
self-inductance (L)	H=Wb/A	`Henry`
magnetic flux (Φ)	Wb=V s	`Weber`
electric flux (Ψ_D)	C=A s	`Coulomb`
permittivity (ϵ)	$F/m=N\ V^{-2}$	`Farad/Meter`
permeability (μ)	$H/m=N\ A^{-2}$	`Henry/Meter`
magnetic induction (**B**)	$T=Wb/m^2$	`Tesla`
magnetic field (**H**)	A/m	`Ampere/Meter`
electric field (**E**)	V/m =N/C	`Volt/Meter`
electric displacement (**D**)	$C\ m^{-2}$	`Coulomb Meter^- 2`
electric current density (**J**)	$A\ m^{-2}$	`Ampere Meter^-2`
magnetic vector potential (**A**)	T m	`Tesla Meter`
Pointing vector (**S**)	$J\ m^{-2}\ s^{-1}$	`Joule Meter^-2 Second^-1`
magnetic dipole moment (m, μ)	$J\ T^{-1}=A\ m^2$	`Joule/Tesla`
electric dipole moment (p_e,d)	C m	`Coulomb Meter`
entropy(S)	$J\ K^{-1}$	`Joule/Kelvin`
Boltzmann constant (k)	$J\ K^{-1}$	`Joule/Kelvin`
Stefan constant (σ)	$W\ m^{-2}\ K^{-4}$	`Watt Meter^-2 Kelvin^-4`
Avogadro constant (N_A)	mol^{-1}	`1/Mole`
Loschmidt constant (n_0)	m^{-3}	`Meter^-3`
molar mass (M)	$kg\ mol^{-1}$	`Kilogram/Mole`

molar volume (V_m)	$m^3\ mol^{-1}$	Meter^3/Mole
Faraday constant (F)	$C\ mol^{-1}$	Coulomb/Mole
molar gas constant (R)	$J\ mol^{-1}\ K^{-1}$	Joule/(Mole Kelvin)
luminous flux (Φ_v)	lm=cd sr	Lumen
illuminance (E_v)	$lx=lm/m^2$	Lux
activity (A)	$Bq=s^{-1}$	Becquerel
absorbed dose (D)	$Gy=J/kg=m^2\ s^{-2}$	Gray
absorbed dose rate ($\dot{D}$)	$Gy\ s^{-1}$	Gray/Second
dose equivalent (H)	$Sv=J/kg=m^2\ s^{-2}$	Sievert
dose equivalent rate ($\dot{H}$)	$Sv\ s^{-1}$	Sievert/Second
exposure (X)	$C\ kg^{-1}$	Coulomb/Kilogram
exposure rate ($\dot{X}$)	$C\ kg^{-1}\ s^{-1}$	Coulomb Kilogram^-1 Second^-1
exposure rate constant (Γ)	$Bq^{-1}\ m^2$	Becquerel^-1 Meter^2
	$C\ kg^{-1}\ s^{-1}$	Coulomb Kilogram^-1 Second^-1
fluence (Φ)	m^{-2}	Meter^-2
fluence rate ($\dot{\Phi}$)	$m^{-2}\ s^{-1}$	Meter^-2 Second^-1
radiation chemical yield (G)	$mol\ J^{-1}$	Mole/Joule

5 Implementation

```
(* :Title: Diana *)

(* :Authors: Rudolf Muradian & Alexander Urintsev,
           Joint Institute for Nuclear Research,
           RU - 141 980, Dubna, Moscow Region, Russia. *)

(* :Summary: The Diana package is intended for making
           dimensional analysis with Mathematica. *)
```



```
(* :Context: Diana` *)

(* :Package Version: 1.0 *)

(* :Keywords: Dimensional analysis *)

(* :Source: H. Langhaar, Dimensional Analysis and Theory
           of Models, Wiley, New York, 1951. *)

(* :Mathematica Version: 2.2 for MS-Windows *)

(* :Warning: Makes use of system symbol Second. *)

(* :Limitation: Do not use the system symbol I as a name in
              input data for the Diana package. *)

BeginPackage["Diana`"]

Print["\n Diana:  Version 1.0, March 1994 \n
        by R. Muradian (1) & A. Urintsev (2) \n
        (1) Electronic address:  muradian@theor.jinrc.dubna.su \n
        (2) Electronic address:  urintsev@sunse.jinr.dubna.su \n
        Type ?Diana`*  for all exported symbols and \n
        ?Diana,  ?DIANA,  ?toFU  for help on the package.", " "]

(********************** usage messages **********************)
Diana::"usage" =
"Diana[{{name1, SIunit1}, {name2, SIunit2}, ...}] performs \n
dimensional analysis and expresses outcome in the form:  \n
name1 -> expr F(K2, K3, ..., Kn).  Here K1, K2, ..., Kn \n
(with K1 = name1/expr) represent the complete set of \n
```

```
dimensionless parameters and F is undetermined function. \n
Parameters K2, ..., Kn and expr do not depend on name1. \n
If n = 1, the result is  name1 -> expr."

DIANA::"usage" =
"DIANA[{{name1, SIunit1}, {name2, SIunit2}, ...}] performs \n
dimensional analysis and expresses outcome in the form: \n
G(K1, K2, ..., Kn) -> 0, where Ki represents the complete set \n
of dimensionless parameters and G is undetermined function."

toFU::"usage" =
"toFU[SIunit] transforms any combination of fundamental and \n
derived SI units into fundamental SI units."

F::"usage" =
   "F is undetermined function of n-1 dimensionless variables."
G::"usage" =
   "G is undetermined function of n dimensionless variables ."


(**************************** toFU ****************************)
toFU[Diana'Private'x_] := (Diana'Private'x /.
       {Radian -> 1,
        Steradian -> 1,
        Newton -> (Meter Kilogram)/Second^2,
        Pascal -> Kilogram/(Meter Second^2),
        Joule -> (Meter^2 Kilogram)/Second^2,
        Watt -> (Meter^2 Kilogram)/Second^3,
        Coulomb -> Ampere Second,
        Volt -> (Meter^2 Kilogram)/(Second^3 Ampere),
        Farad -> (Second^4 Ampere^2)/(Meter^2 Kilogram),
        Ohm -> (Meter^2 Kilogram)/(Second^3 Ampere^2),
```



```
        Siemens -> (Second^3 Ampere^2)/(Meter^2 Kilogram),
        Weber -> (Meter^2 Kilogram)/(Second^2 Ampere),
        Tesla -> Kilogram/(Second^2 Ampere),
        Henry -> (Meter^2 Kilogram)/(Second^2 Ampere^2),
        Lumen -> Candela Steradian,
        Lux -> (Candela Steradian)/Meter^2,
        Hertz -> 1/Second,
        Becquerel -> 1/Second,
        Gray -> Meter^2/Second^2,
        Sievert -> Meter^2/Second^2}) //.
        {(Diana'Private'a_^Diana'Private'p_)^Diana'Private'q_ :>
         Diana'Private'a^(Diana'Private'p Diana'Private'q),
         (Diana'Private'c_ Diana'Private'd_)^Diana'Private'r_ :>
         Diana'Private'c^Diana'Private'r *
         Diana'Private'd^Diana'Private'r}

(***************** fundamental SI units usage ****************)
Map[(Evaluate[First[#]]::"usage" =
    StringJoin[ToString[ First[#] ],
               " is the fundamental SI unit of ",
               ToString[ Last[#]]
     ]
    ) &,
    {{Meter, "length."},{Kilogram, "mass."},{Second, "time."},
     {Ampere, "electric current."},
     {Kelvin, "thermodynamic temperature."},
     {Mole, "amount of substance."},
     {Candela, "luminous intensity."},
     {$, "price (dollar), according \n
to the proposed extension of the SI units."}
    }
   ]
```

```
$NewMessage[Second, "usage"]
Second::"usage" = StringJoin[Second::"usage",
" It is also used as the fundamental SI unit of time."]

(******************** derived SI units usage ****************)
Radian::"usage" =
   "Radian is a dimensionless measure of plane angle."
Steradian::"usage" =
   "Steradian is a dimensionless measure of solid angle."

Map[(Evaluate[First[#]]::"usage" =
    StringJoin[" ", ToString[ First[#] ],
               " is the derived SI unit of ",
                ToString[ Last[#]],
                "\n ", ToString[ First[#] ], " = ",
                ToString[ InputForm[Evaluate[toFU[First[#]]]]]
              ]
     ) &,
      {{Newton, "force."}, {Pascal, "pressure."},
       {Joule, "energy."}, {Watt, "power."},
       {Coulomb, "electric charge."},
       {Volt, "electric potential difference."},
       {Farad, "capacitance."}, {Ohm, "electric resistance."},
       {Siemens, "electric conductance."},
       {Weber, "magnetic flux."},
       {Tesla, "magnetic flux density (induction)."},
       {Henry, "inductance."},
       {Lumen, "luminous flux."},
       {Lux, "illuminance (illumination)."},
       {Hertz, "frequency."}, {Becquerel, "radioactivity."},
       {Gray, "absorbed dose of radiation."},
```



```
        {Sievert, "dose equivalent."}
      }
   ]

Begin["Diana`Private`"]

(********************** fundamental SI units *****************)
FUnits = {Meter, Kilogram, Second, Ampere, Kelvin, Mole,
          Candela, $}

(************************** SI units ************************)
SIunits =  {$, Ampere, Becquerel, Candela, Coulomb, Farad, Gray,
            Henry, Hertz, Joule, Kelvin, Kilogram, Lumen, Lux,
            Meter, Mole, Newton, Ohm, Pascal, Radian, Second,
            Siemens, Sievert, Steradian, Tesla, Volt, Watt,
            Weber}

(********************** checking units **********************)
checkUnits[n_, expr_] := Module[{i, wd = False},
  Do[If[UnsameQ[Complement[Variables[expr[[i, 2]]], SIunits],
        {}], wd = True; Break[]], {i, n}]; wd]

(********************* warning messages ********************)
Diana::"forbidden name" = "Symbol I is forbidden for use."
Diana::"wrong data" =
   "Problem cannot be solved. Check input for Diana."
Diana::"wrong units" = "Input units for Diana are not SI units."
DIANA::"forbidden name" = "Symbol I is forbidden for use."
DIANA::"wrong data" =
   "Problem cannot be solved. Check input for DIANA."
DIANA::"wrong units" = "Input units for DIANA are not SI units."
```

```
(************************ exponent *************************)
exponent[expr_, x_] :=
   If[FreeQ[expr, x], 0, expr /. _. x^d_. :> Rationalize[d]]

(*************************** Diana *************************)
Diana[expr:{{_, _}, {_, _} ...}] :=
Module[{a, b, i, j, k, m}, If[Not[FreeQ[expr, I]],
          Return[Message[Diana::"forbidden name"]]];
   k = Length[expr]; If[checkUnits[k, expr],
                        Return[Message[Diana::"wrong units"]]];
   m = Length[FUnits];
   a = Table[exponent[toFU[expr[[i, 2]]], FUnits[[j]]],
                 {j, m}, {i, k}];
   b = a[[Range[m], Range[2, k]]];
   Check[a = Together[LinearSolve[b, Map[#[[1]]&, a]]],
         Return[Message[Diana::"wrong data"]]];
   b = NullSpace[b];  m = k - 1;
   expr[[1, 1]] -> Product[expr[[j + 1, 1]]^a[[j]], {j, m}] *
       If[SameQ[b, {}], 1,
          F @@ Table[Product[expr[[j + 1, 1]]^b[[i, j]],
                 {j, m}], {i, Length[b]}]]
]

(*************************** DIANA *************************)
DIANA[expr:{{_, _}, {_, _} ...}] :=
Module[{a, i, j, k},
       If[Not[FreeQ[expr, I]],
          Return[Message[DIANA::"forbidden name"]]];
       k = Length[expr];
       If[checkUnits[k, expr],
          Return[Message[DIANA::"wrong units"]]];
       a = NullSpace[Table[exponent[toFU[expr[[i, 2]]],
```



```
                         FUnits[[j]]], {j, Length[FUnits]}, {i, k}]];
       If[SameQ[a, {}], Return[Message[DIANA::"wrong data"]]];
       G @@ Table[Product[expr[[j, 1]]^a[[i, j]], {j, k}],
             {i, Length[a]}] -> 0
]

End[] (* Diana`Private` *)
EndPackage[] (* Diana *)
```

6 Finish

Computer experimentation in the area of dimensional analysis can help for better understanding of the old problems and give impact into deeper philosophic insight into new research fields. *Diana* can assist researchers in a broad field of physics, physical chemistry, engineering and perhaps economics. We hope that *Diana* will help you to open the door into marvelous world of the *Mathematica*®1 Physics.

The file *Diana*.ma in *Mathematica* Notebook format is in preparation.

References

[1] S. Wolfram: *Mathematica: A System for Doing Mathematics by Computer*, 2nd edition, Addison-Wesley, Redwood City, CA, 1991.

[2] P. Boyland et al.: *Guide to Standard Mathematica Packages*, Technical Report, Wolfram Research, 1991.

[3] H. Langhaar: *Dimensional Analysis and Theory of Models* , Wiley, New York, 1951.

[4] C. Focken: *Dimensional Methods and Their Applications*, E. Arnold, London, 1953.

[5] L. Sedov: *Similarity and Dimensional Methods in Mechanics*, Academic Press, New York, 1959.

[6] G. Taylor: *The formation of a Blast Wave by a Very Intense Explosion*, Proceedings of the Royal Society, **A 201**, 159, 1950.

[7] W. Remillard: *Applying Dimensional Analysis*, Am. J. Phys. **51**, 137, 1983.

[8] V. Matveev, R. Muradian and A. Tavkhelidze: *Automodellism in the Large- Angle Elastic Scattering and Structure of Hadrons*, Letters Nuovo Cimento, **7**, 719, 1973.

[9] G. West: *Scale and Dimension—From Animals to Quarks*, Los Alamos Science **11**, 2, 1984.

[10] Z. Rácz: *Examples of Unconventional Dimensional Analysis*, Acta Physica Hungarica **72**, 249, 1992.

[11] M.S. Quaraishi and T.Z. Fahidy: *A Simplified Procedure For Dimensional Analysis Employing SI Units* , The Canadian Journal of Chemical Engineering, **59**, 563, 1981.

[12] T.Z. Fahidy and M.S. Quaraishi: *Principles of Dimensional Analysis* , In: Encyclopedia of Fluid Mechanics, Chapter 12, p. 400. Gulf Publishing Company, Houston, TX, 1986.

[13] E. Cohen and P. Giacomo: *Symbols, Units, Nomenclature and Fundamental Constants in Physics*, Physica **A 146**, 1, 1987.

[14] E. Cohen and B. Taylor: *The 1986 Adjustment of the Fundamental Physical Constants*, Rev. Mod. Phys. **59**, 1121, 1987.

[15] *SI Units in Radiation Protection and Measurements*, National Council on Radiation Protection and Measurements, report No 82, Bethesda, MD, 1985.

Received by Publishing Department
on March 30, 1994.